
Fig. 1. Plots of a sharp vMF distribution (𝜅 = 10000000, 𝛍 = [0, 0, 1]) with the traditional form (a, b) and our
form (c). The horizontal axis is the angle between a direction 𝛚 and the vMF center axis 𝛍. (b) The traditional
vMF form with single precision produces a significant numerical error. (c) Our vMF form is more numerically
stable than the traditional form.

1 Introduction
The von Mises–Fisher (vMF) distribution [1953] on S2 is a normalized spherical Gaussian defined as

𝑝 (𝛚; 𝛍, 𝜅) = 𝜅

4𝜋 sinh𝜅 exp(𝜅 (𝛚 · 𝛍)), (1)

where 𝛚 ∈ S2 is a unit vector, 𝛍 ∈ S2 is the center axis of the vMF distribution, and 𝜅 ∈ [0,∞) is
the sharpness of the vMF distribution. This distribution has often been used in computer graphics,
such as real-time lighting approximation [Tsai and Shih 2006] and path guiding [Dong et al. 2023;
Ruppert et al. 2020]. However, a straightforward implementation of the vMF distribution using
floating points can produce a noticeable numerical error. Therefore, we describe a numerically
stable implementation of the vMF distribution.

2 Numerically Stable Form of the vMF Distribution
Eq. 1 can produce NaN because exp(𝜅 (𝛚 · 𝛍)) and sinh(𝜅) can overflow for large 𝜅 (e.g., 𝜅 >

arsinh((2 − 2−23) × 2127) ≈ 89.4 for single precision). To avoid such NaN, computer graphics
applications have often used the following equivalent form:

𝑝 (𝛚; 𝛍, 𝜅) = 𝜅

2𝜋 (1 − exp(−2𝜅)) exp(𝜅 ((𝛚 · 𝛍) − 1)), (2)

where exp(𝜅 ((𝛚 · 𝛍) − 1)) ∈ (0, 1] is the unnormalized spherical Gaussian. On the other hand, Eq. 2
in floating point can produce a significant error for 𝜅 → 0 and 𝜅 → ∞. Therefore, we use a more
numerically stable form. To improve the stability for small 𝜅, we use an accurate implementation
of 𝑎(𝑥) = 𝑥/(exp(𝑥) − 1) [Higham 2002] for the normalization factor 𝜅

2𝜋 (1−exp(−2𝜅 ) ) as follows:

𝑝 (𝛚; 𝛍, 𝜅) = 𝑎(−2𝜅)
4𝜋 exp(𝜅 ((𝛚 · 𝛍) − 1)) . (3)

For an HLSL implementation of 𝑎(𝑥), please see Listing 1. Although the above form is accurate
for the normalization factor, the unnormalized spherical Gaussian term exp(𝜅 ((𝛚 · 𝛍) − 1)) is
still numerically unstable for 𝛚 → 𝛍. This numerical error can be noticeable for a sharp vMF
distribution (Fig. 1). For applications that require numerical accuracy for such high-frequency
distributions, we use the Euclidean distance between 𝛚 and 𝛍 instead of (𝛚 · 𝛍) − 1 as follows:

𝑝 (𝛚; 𝛍, 𝜅) = 𝑎(−2𝜅)
4𝜋 exp

(
−𝜅2 ∥𝛚 − 𝛍∥2

)
. (4)

Listing 2 shows our vMF implementation using the above form.

Listing 1. 𝑎(𝑥) = 𝑥/(exp(𝑥) − 1) with cancellation of rounding errors [Higham 2002] (HLSL).
float x_over_expm1(float x) {
float u = exp(x);
if (u == 1.0f) { return 1.0f; }
float y = u - 1.0f;
if (abs(x) < 1.0f) { return log(u) / y; }
return x / y;

}

Listing 2. Our numerically stable vMF implementation (HLSL). Instead of using 𝛚 · 𝛍, we use the Euclidean
distance between 𝛚 and 𝛍.
float vmf(float3 dir , float3 axis , float sharpness) {
float3 d = dir - axis;
return exp(-0.5f * sharpness * dot(d, d)) * x_over_expm1 (-2.0f * sharpness) / (4.0f * M_PI);

}

3 Sampling of the vMF Distribution
To sample a direction 𝛚 according to the vMF distribution 𝑝 (𝛚; 𝛍, 𝜅), we first sample a direction
[cos𝜙 sin𝜃, cos𝜙 sin𝜃, cos𝜃 ] ∈ S2 in a local frame, where 𝜃 ∈ [0, 𝜋] and 𝜙 ∈ [0, 2𝜋) are the polar
coordinates of this local direction. Then, we rotate the local direction into world space. For this
case, the azimuthal angle 𝜙 is uniformly distributed as follows:

𝜙 = 2𝜋𝜉0, (5)

where 𝜉0 ∈ [0, 1) is a uniform random number. To sample cos𝜃 = 𝛚 · 𝛍 using a different uniform
random number 𝜉1 ∈ [0, 1), Jakob [2012] improved the numerical stability from Jung [2009] by
deriving the following form:

cos𝜃 = 1 + 1
𝜅
log (𝜉1 + (1 − 𝜉1) exp(−2𝜅)) . (6)

However, this sampling can still produce a significant error for small 𝜅 , because the precision of the
random variable is lost by 𝜉1 + (1 − 𝜉1) exp(−2𝜅) → 1 for 𝜅 → 0. To reduce the error, we replace 𝜉1
with 1 − 𝜉1 in Eq. 6 as follows:

cos𝜃 = 1 + 1
𝜅
log (1 − 𝜉1 + 𝜉1 exp(−2𝜅)) = 1 + 1

𝜅
log1p (𝜉1expm1(−2𝜅)) , (7)

where expm1(𝑥) = exp(𝑥) − 1 and log1p(𝑥) = log(1 + 𝑥) are built-in functions available in some
programming languages (e.g., C++), and they are numerically stable for small |𝑥 |. When 𝜅 is small,
|𝜉1expm1(−2𝜅) | is small. Therefore, Eq. 7 reduces the numerical error for small 𝜅. The same form
was used by Frisch and Hanebeck [2023] for their deterministic sampling.

Fig. 2. Plots of sample directions [cos𝜙 sin𝜃, sin𝜙 sin𝜃 ] in the local frame for vMF distributions. For low-
frequency distribution (upper row) and high-frequency distribution (lower row), Jakob [2012]’s method with
single precision (b) generates highly correlated samples due to numerical errors, while ours (c) does not.

Once we get cos𝜃 , we then calculate sin𝜃 . Although Jakob [2012] used sin𝜃 =
√
1 − cos2 𝜃 , it

can produce a noticeable error due to catastrophic cancellation when cos𝜃 → 1. To avoid the
catastrophic cancellation for sin𝜃 , we use the following equation:

𝑟 =

{
1
𝜅
log1p (𝜉1expm1(−2𝜅)) if 𝜅 > 𝑡

−2𝜉1 if 𝜅 ≤ 𝑡
, (8)

cos𝜃 = 1 + 𝑟 , (9)

sin𝜃 =
√
−𝑟 2 − 2𝑟 =

√︁
−fma(𝑟, 𝑟, 2𝑟 ) , (10)

where 𝑡 = 0 for the exact solution, and fma(𝑥,𝑦, 𝑧) = 𝑥 × 𝑦 + 𝑧 is the fused multiply-add operation
to reduce the numerical error in floating-point arithmetic. Even if the built-in fma function is
not available, the calculation of sin𝜃 =

√
−𝑟 2 − 2𝑟 is still more numerically stable than sin𝜃 =√

1 − cos2 𝜃 for cos𝜃 → 1. For a sharp distribution with large𝜅 , 𝑟 is densely and precisely distributed
around zero. Therefore, Eq. 10 produces accurate sin𝜃 around zero. Fig. 2 shows plots of samples
generated using our method. Listing 3 shows an HLSL implementation for our sampling routine.

To further improve the numerical stability, we use 𝑡 = 𝜖/4 where 𝜖 is the machine epsilon. This
is because, let fl(𝑓 (·)) be an operation 𝑓 (·) in floating-point arithmetic, and 𝑥 be a floating point
value, then fl(expm1(𝑥)) = 𝑥 and fl(log1p(𝑥)) = 𝑥 when |𝑥 | ≤ 𝜖/2. Therefore, for floating-point 𝜅

and 𝜉1, we obtain

𝑟 ≈ fl
(
fl (log1p (𝜉1fl(expm1(−2𝜅))))

𝜅

)
= fl

(
−2𝜅𝜉1
𝜅

)
≈ −2𝜉1 for 0 < 𝜅 ≤ 𝜖/4. (11)

The rightmost approximation 𝑟 ≈ −2𝜉1 is more accurate than calculating the exact form in floating-
point arithmetic.

Listing 3. Numerically stable sampling of the vMF distribution. Since HLSL does not have a built-in fma
function for single precision, we use the mad function instead. For the implementation details of expm1, log1p,
and orthonormal_basis functions, please see Listings 4, 5, and 6, respectively.
float3 sample_vmf(float2 rand , float3 axis , float sharpness) {
float phi = 2.0f * M_PI * rand.x;
float THRESHOLD = FLT_EPSILON / 4.0f;
float r = sharpness > THRESHOLD ? log1p(rand.y * expm1 (-2.0f * sharpness)) / sharpness

: -2.0f * rand.y;
float cos_theta = 1.0f + r;
float sin_theta = sqrt(-mad(r, r, 2.0f * r));
float3 dir = {cos(phi) * sin_theta , sin(phi) * sin_theta , cos_theta };
float3x3 frame = orthonormal_basis(axis);
return mul(dir , frame);

}

Listing 4. expm1(𝑥) = exp(𝑥) − 1 with cancellation of rounding errors [Higham 2002] (HLSL). Since HLSL
does not have a built-in expm1 function unlike C++, we use this implementation as a workaround.
float expm1(float x) {
float u = exp(x);
if (u == 1.0f) { return x; }
float y = u - 1.0f;
if (abs(x) < 1.0f) { return x * y / log(u); }
return y;

}

Listing 5. log1p(𝑥) = log(𝑥 + 1) with cancellation of rounding errors [Goldberg 1991] (HLSL). Since HLSL
does not have a built-in log1p function unlike C++, we use this implementation as a workaround. For this
classic algorithm, aggressive compiler optimization must be disabled for floating points.
float log1p(float x) {
// For this algorithm , we must prevent compilers from optimizing (x + 1) - 1 to x.
volatile float u = x + 1.0f;
if (u == 1.0f) { return x; }
float y = log(u);
if (x < 1.0f) { return x * y / (u - 1.0f); }
return y;

}

Listing 6. Building of an orthonormal basis [Duff et al. 2017] (HLSL). We use this basis for the local frame of
the vMF distribution.
float3x3 orthonormal_basis(float3 axis) {
float s = axis.z >= 0.0f ? 1.0f : -1.0f;
float c = -1.0f / (s + axis.z);
float b = axis.x * axis.y * c;
float3 b1 = {1.0f + s * axis.x * axis.x * c, s * b, -s * axis.x};
float3 b2 = {b, s + axis.y * axis.y * c, -axis.y};
return float3x3(b1 , b2, axis);

}

